

# Применение функции ИНДЕКС

# ИНДЕКС. Вариант 1. Извлечение данных из столбца по номеру ячейки

**Задача:** нужно извлечь данные из одномерного диапазона-столбца, если мы знаем порядковый номер ячейки. Синтаксис в этом случае будет:


E2		✕ ✓ f <sub>x</sub>		=ИНДЕКС(A2:A7;3)		
	A	B	C	D	E	F
1	Список					
2	Москва		Третий по счету город		Лондон	
3	Пекин					
4	Лондон					
5	Париж					
6	Берлин					
7	Будапешт					

**=ИНДЕКС(Диапазон\_столбец; Порядковый\_номер\_ячейки)**

# То же с функцией ПОИСКПОЗ

Функция **ИНДЕКС** часто используется в связке с функцией **ПОИСКПОЗ(MATCH)**, которая выдает номер искомого значения в диапазоне.

Таким образом, эта пара заменяет **ВПР...**



The screenshot shows an Excel spreadsheet with a formula bar at the top displaying the formula `=ИНДЕКС(A2:A7;ПОИСКПОЗ(D2;B2:B7;0))`. The spreadsheet contains a table with two columns: 'Страна' (Country) and 'Город' (City). The data is as follows:

	А	В	С	Д	Е	Ф	Г	Н
1	Страна	Город						
2	Россия	Москва		Париж		Франция		
3	Китай	Пекин						
4	Англия	Лондон						
5	Франция	Париж						
6	Германия	Берлин						
7	Венгрия	Будапешт						

... но, в отличие от ВПР, могут извлекать значения левее поискового столбца и номер столбца-результата высчитывать не нужно.

# Функция ПОИСКПОЗ

Основное назначение этой функции в том, чтобы искать позицию заданного элемента в наборе значений. Чаще всего она применяется для поиска порядкового номера ячейки в диапазоне, где лежит нужное нам значение.

Синтаксис этой функции следующий:

**=ПОИСКПОЗ(Что\_ищем; Где\_ищем; Режим\_поиска)**

где

- **Что\_ищем** - это значение, которое надо найти
- **Где\_ищем** - это одномерный диапазон или массив (строка или столбец), где производится поиск
- **Режим\_поиска** - как мы ищем: точно (0), с округлением в большую сторону (-1) или в меньшую сторону (1)

# ПОИСКПОЗ: Точный поиск

Классический сценарий - поиск точного текстового совпадения для нахождения позиции нужного текста или числа в списке:

F2				fx		=ПОИСКПОЗ("Даниил";A1:A10;0)	
	A	B	C	D	E	F	G
1	Иван						
2	Михаил		Имя "Даниил" в списке по счету			7	
3	Ольга						
4	Сергей						
5	Юлия						
6	Никита						
7	Даниил						
8	Мария						
9	Егор						
10	Кирилл						

# ПОИСКПОЗ: Поиск первой или последней текстовой ячейки

Если в качестве искомого значения задать звездочку, то функция будет искать *первую ячейку* с текстом и выдавать её позицию.

Для поиска *последней текстовой ячейки* можно изменить третий аргумент **Режим\_поиска** с нуля на минус 1. Числа и пустые ячейки в этом случае игнорируются.

N19										
	A	B	C	D	E	F	G	H	I	
1										
2	12	Номер первой текстовой ячейки				5	=ПОИСКПОЗ("*";A1:A10;0)			
3										
4	45	Номер последней текстовой ячейки				7	=ПОИСКПОЗ("*";A1:A10;-1)			
5	Иван									
6	99									
7	Мария									
8	23									
9										
10	55									
11										

# ПОИСКПОЗ: Поиск ближайшего числа или даты (1)

Если последний аргумент задать равным 1 или -1, то можно реализовать поиск *ближайшего наименьшего или наибольшего* числа. Таблица при этом обязательно должна быть отсортирована по возрастанию или убыванию соответственно. Это похоже на интервальный просмотр у функции ВПР (VLOOKUP), но там возможен только поиск ближайшего наименьшего, а здесь - есть выбор.

Ищем *ближайшую наименьшую* по мощности модель (последний аргумент задать равным 1 и отсортировать таблицу по возрастанию):

G4							
=ПОИСКПОЗ(G2;B2:B6;1)							
	A	B	C	D	E	F	G
	Модель генератора	Номинальная мощность, кВт					
1							
2	Пушинка	9			Требуется, кВт		47
3	Нормаль	12					
4	Зверь	35			Нужна модель №		3
5	Бомба	49					
6	Днепрогэс	65					
7							

# ПОИСКПОЗ: Поиск ближайшего числа или даты (2)

Если третий аргумент равен -1 и таблица отсортирована по убыванию, то мы найдем ближайшую более мощную модель.

G4		✕ ✓ <i>fx</i>		=ПОИСКПОЗ(G2;B2:B6;-1)			
	A	B	C	D	E	F	G
1	Модель генератора	Номинальная мощность, кВт					
2	Днепрогэс	65			Требуется, кВт		47
3	Бомба	49					
4	Зверь	35			Нужна модель №		2
5	Нормаль	12					
6	Пушинка	9					



# ПОИСКПОЗ: Связка функций ПОИСКПОЗ и ИНДЕКС (повторение)

**Задача:** получить не номер, а название модели генератора.

G4		✕ ✓ f <sub>x</sub>		=ПОИСКПОЗ(G2;B2:B6;-1)			
	A	B	C	D	E	F	G
1	Модель генератора	Номинальная мощность, кВт					
2	Днепрогэс	65			Требуется, кВт		47
3	Бомба	49					
4	Зверь	35			Нужна модель №		2
5	Нормаль	12					
6	Пушинка	9					

H4		✕ ✓ f <sub>x</sub>		=ИНДЕКС(A2:A5;G4)				
	A	B	C	D	E	F	G	H
	Модель генератора	Номинальная мощность, кВт						
1								
2	Днепрогэс	<div></div> 65			Требуется, кВт		47	
3	Бомба	<div></div> 49						
4	Зверь	<div></div> 35			Нужна модель №		2	Бомба
5	Нормаль	<div></div> 12						
6	Пушинка	<div></div> 9						

# ПОИСКПОЗ: Связка функций ПОИСКПОЗ и ИНДЕКС (с датами)

**Задача:** определить, на каком этапе сейчас находится проект.

F4	=ПОИСКПОЗ(F2;B2:B8;1)						
	A	B	C	D	E	F	G
1	Название этапа	Начало этапа					
2	Сбор информации	01.03.2017			Сегодня	13.03.2017	
3	Мозговой штурм	07.03.2017					
4	Отбор идей	11.03.2017		Мы на этапе		3	Отбор идей
5	Пробная реализация	18.03.2017					
6	Внедрение в фокус-группах	26.03.2017					
7	Сбор обратной связи	31.03.2017					
8	Корректировки макета	08.04.2017					

G4	=ИНДЕКС(A2:A8;F4)						
	A	B	C	D	E	F	G
	Название этапа	Начало этапа					
	Сбор информации	01.03.2017			Сегодня	13.03.2017	
	Мозговой штурм	07.03.2017					
	Отбор идей	11.03.2017		Мы на этапе		3	Отбор идей
	Пробная реализация	18.03.2017					
	Внедрение в фокус-группах	26.03.2017					
	Сбор обратной связи	31.03.2017					
	Корректировки макета	08.04.2017					

# ПОИСКПОЗ: ПОИСКПОЗ и ИНДЕКС (двумерный поиск)

Используем сразу два **ПОИСКПОЗ**а, вложенных в **ИНДЕКС**, чтобы реализовать двумерный поиск по строке и столбцу одновременно.

Н6          =ИНДЕКС(B2:E10;ПОИСКПОЗ(Н4;A2:A10;0);ПОИСКПОЗ(Н2;B1:E1;0))										
	A	B	C	D	E	F	G	H	I	J
1		Москва	Самара	Казань	Воронеж					
2	янв	688	674	301	421		Город	Казань		
3	фев	609	829	607	632					
4	мар	566	629	787	81		Месяц	июн		
5	апр	152	174	90	343					
6	май	457	151	67	243		Значение	287		
7	июн	438	290	287	601					
8	июл	138	898	22	951					
9	авг	478	286	938	969					
10	сен	621	321	883	809					

# ИНДЕКС. Вариант 2. Извлечение данных из двумерного диапазона

Если диапазон двумерный, т.е. состоит из нескольких строк и столбцов, то функция ИНДЕКС будет использоваться немного в другом формате, т.е. функция извлекает значение из ячейки диапазона с пересечения строки и столбца с заданными номерами:

**=ИНДЕКС(Диапазон; Номер\_строки; Номер\_столбца)**

J2										
	A	B	C	D	E	F	G	H	I	J
1	Список	янв	фев	мар						
2	Москва	5671	9332	4273		Значение у третьего города во втором месяце				8012
3	Пекин	6057	7756	882						
4	Лондон	8704	8012	8650						
5	Париж	418	5938	1280						
6	Берлин	6424	6856	9853						
7	Будапешт	2363	7203	4940						
8										

# ИНДЕКС. Вариант 2. Извлечение данных из двумерного диапазона (+ две ПОИСКПОЗ)

Двумерный поиск:

J2													
	A	B	C	D	E	F	G	H	I	J	K	L	
1	Список	янв	фев	мар									
2	Москва	5671	9332	4273		Париж		мар		1280			
3	Пекин	6057	7756	882									
4	Лондон	8704	8012	8650									
5	Париж	418	5938	1280									
6	Берлин	6424	6856	9853									
7	Будапешт	2363	7203	4940									
8													



## ИНДЕКС. Вариант 3. Несколько таблиц

Если таблица не одна, а их несколько, то функция **ИНДЕКС** может извлечь данные из нужной строки и столбца именно заданной таблицы. В этом случае используется следующий синтаксис:

**=ИНДЕКС((Диапазон1;Диапазон2;Диапазон3); Номер\_строки;  
Номер\_столбца; Номер\_диапазона)**

J10														
	A	B	C	D	E	F	G	H	I	J	K	L	M	N
1		янв	фев	мар			янв	фев	мар			янв	фев	мар
2	Москва	5671	9332	4273		Москва	4509	5006	1933		Москва	307	3950	7333
3	Пекин	6057	7756	882		Пекин	5273	4760	2156		Пекин	8629	7289	3038
4	Лондон	8704	8012	8650		Лондон	9275	9376	6578		Лондон	2472	8493	4337
5	Париж	418	5938	1280		Париж	6833	7355	1298		Париж	4462	2988	5082
6	Берлин	6424	6856	9853		Берлин	4813	2607	1089		Берлин	2658	7734	968
7	Будапешт	2363	7203	4940		Будапешт	9424	4953	1144		Будапешт	256	2293	5853
8														
9														
10														
11														

Обратите особое внимание, что в этом случае первый аргумент – список диапазонов - заключается **в скобки**, а сами диапазоны перечисляются через **точку с запятой**.

## ИНДЕКС. Вариант 4. Ссылка на столбец / строку

Если во втором варианте использования функции ИНДЕКС номер строки или столбца задать **равным нулю (или просто не указать)**, то функция будет выдавать уже не значение, а ссылку на диапазон-столбец или диапазон-строку соответственно.

	A	B	C	D	E	F	G	H	I	J	K	L
1	Список	янв	фев	мар								
2	Москва	5671	9332	4273	Сумма всех значений по третьему городу				25366	=СУММ(ИНДЕКС(B2:D7;3;0))		
3	Пекин	6057	7756	882								
4	Лондон	8704	8012	8650								
5	Париж	418	5938	1280								
6	Берлин	6424	6856	9853								
7	Будапешт	2363	7203	4940								
8												
9												
10	Список	янв	фев	мар								
11	Москва	5671	9332	4273	Сумма всех значений за февраль				45097	=СУММ(ИНДЕКС(B11:D16;0;2))		
12	Пекин	6057	7756	882								
13	Лондон	8704	8012	8650								
14	Париж	418	5938	1280								
15	Берлин	6424	6856	9853								
16	Будапешт	2363	7203	4940								

Обратите внимание, что поскольку **ИНДЕКС** выдает в этом варианте не конкретное значение ячейки, а **ссылку на диапазон**, то для подсчета потребуется заключить ее в дополнительную функцию, например **СУММ (SUM)**, **СРЗНАЧ (AVERAGE)** и т.п.

## ИНДЕКС. Вариант 5. Ссылка на ячейку

Если взять функцию ИНДЕКС в первом или втором варианте и подставить ее после двоеточия, то наша функция будет выдавать уже не значение, а адрес, и на выходе мы получим полноценную ссылку на диапазон от начальной ячейки до той, которую нашла ИНДЕКС:

H2		✕ ✓ fx		=СУММ(B2:ИНДЕКС(B2:B7;F2))				
	A	B	C	D	E	F	G	H
1	Город	Продажи						
2	Москва	6050		Продажи первых		3	городов	17543
3	Пекин	1821						
4	Лондон	9672						
5	Париж	1254						
6	Берлин	9252						
7	Будапешт	5158						

Нечто похожее можно реализовать функцией СМЕЩ (OFFSET), но она, в отличие от ИНДЕКС, является волатильной, т.е. пересчитывается каждый раз при изменении любой ячейки листа. ИНДЕКС же работает более тонко и запускает пересчет только при изменении своих аргументов, что ощутимо ускоряет расчет в тяжелых книгах по сравнению со СМЕЩ.